
PhoenixBricks
A set of proposed patterns to improve code organization for Phoenix application
Installation
To install PhoenixBricks, add it to your list of dependencies in mix.exs.
def deps do
  [
    {:phoenix_bricks, "~> 0.3.0"}
  ]
end
Once you've added PhoenixBricks to your list, update your dependencies by running:
$ mix deps.get
Getting started
PhoenixBricks it is intented as a set of proposed pattern to organize code into a Phoenix application in a modular way.
Queries
A query is a module that receives a list of atom/keywords and returns an Ecto.Query improved with provided scopes
ProductQuery.scope([title_matches: "a value", price_is_greater_than: 1000])
=> #Ecto.Query<from p0 in PhoenixBricks.Catalogue.Product,
 where: ilike(p0.title, ^"%a value%"), where: p0.price >= ^1000
The idea is to have a query composer that could be easy to use like ActiveRecord.
To generate a query for the Catalogue.Product schema with some additional custom scopes, simply run
$ mix phx.bricks.gen.query Catalogue.Product name:matches:string price:lte:integer
The generated module will contain a set of default scopes for simple column matchers and the additional scopes provided in the command.
defmodule PhoenixBricks.Catalogue.ProductQuery do
  @moduledoc false

  import Ecto.Query, warn: false

  def starting_scope do
    PhoenixBricks.Catalogue.Product
  end

  def scope(scopes, starting_scope) do
    scopes
    |> Enum.reduce(starting_scope, fn scope, query ->
      apply_scope(query, scope)
    end)
  end

  def scope(scopes) do
    scope(scopes, starting_scope())
  end

  def scope do
    starting_scope()
  end

  defp apply_scope(query, {column, {:eq, value}}) do
    where(query, [q], field(q, ^column) == ^value)
  end

  defp apply_scope(query, {column, {:neq, value}}) do
    where(query, [q], field(q, ^column) != ^value)
  end

  defp apply_scope(query, {column, {:lte, value}}) do
    where(query, [q], field(q, ^column) <= ^value)
  end

  defp apply_scope(query, {column, {:lt, value}}) do
    where(query, [q], field(q, ^column) < ^value)
  end

  defp apply_scope(query, {column, {:gte, value}}) do
    where(query, [q], field(q, ^column) >= ^value)
  end

  defp apply_scope(query, {column, {:gt, value}}) do
    where(query, [q], field(q, ^column) > ^value)
  end

  defp apply_scope(query, {column, {:matches, value}}) do
    value = "%#{value}%"
    where(query, [q], ilike(field(q, ^column), ^value))
  end

  defp apply_scope(query, {:title_matches, value}) do
    apply_scope(query, {:title, {:matches, value}})
  end

  defp apply_scope(query, {:price_gte, value}) do
    apply_scope(query, {:price, {:gte, value}})
  end
end
Using the ProductQuery module it's possible to rewrite the list_products/0 method of the context this way:
defmodule PhoenixBricks.Catalogue do
  ...
  def list_products(scopes \\ []) do
    ProductQuery.scope(scopes)
    |> Repo.all()
  end
  ...
end

[title_matches: "a value"]
|> Catalogue.list_products()
If you wanto to add custom scopes you simply have to add new definitions of the apply_scope/2 method:
defp apply_scope(query, :active) do
  from(q in query, where: q.active == true)
end

defp apply_scope(query, {:some_scope_name, value}) do
  from(query in q, ...)
end
...
Catalogue.list_products([
  :active,
  title_matches: "a value",
  some_scope_name: "some value"
])
Filters
A filter is a module that receives a map of filters (for instance coming from a search form) and returns a list of allowed filters.
%{"filters" => %{"title_matches" => "a value", "not_allowed" => "xxx"}}
|> ProductFilter.params_to_scopes()
=> [title_matches: "a value"]
To generate a module filter for the Catalogue.Product schema you simply have to run
$ mix phx.bricks.gen.filters Catalogue.Product title:matches:string price:gte:integer
that will create a schema that we could use if a form for records filtering.
defmodule CrudExample.Catalogue.ProductFilter do
  @moduledoc false

  use Ecto.Schema

  import Ecto.Changeset

  alias CrudExample.Catalogue.ProductFilter

  embedded_schema do
    field :title_matches, :string
    field :price_gte, :integer
  end

  def changeset(filter, attrs) do
    filter
    |> cast(attrs, [:title_matches, :price_gte])
  end

  def params_to_scopes(params) do
    filters = Map.get(params, "filters", %{})

    filter_changeset = changeset(%ProductFilter{}, filters)

    filter_changeset.changes
    |> Enum.map(fn {name, value} -> {name, value} end)
  end
end
# lib/phoenix_bricks/catalogue.ex
def filter_products(params) do
  ProductFilter.changeset(%ProductFilter{}, params)
end


# lib/phoenix_bricks_web/controllers/products_controller.ex
def index(conn, params) do
  filter_changeset =
    ProductFilter.changeset(%ProductFilter{}, params)

  products =
    params
    |> ProductFilter.params_to_scopes()
    |> ProductsQuery.scope()
    |> Repo.all()

    # or
    # params
    # |> ProductsFilter.params_to_scopes()
    # |> Products.list_products()

  render(
    conn,
    "index.html",
    products: products,
    filter_changeset: filter_changeset
  )
end


# lib/phoenix_bricks_web/templates/products/index.html.eex
<%= form_for @filter_changeset, Routes.product_index_path(@conn, :index) method: :get do %>
  <%= label :title_matches %>
  <%= text_input :title_matches %>
<% end %>

<%= for product <- @products do %>
  ...
<% end %>
Query e Filters
The 2 modules can be used in pipeline since the output format for filters are the input format of queries.
%{"filters" => %{"title_matches" => "title", "price_gte" => "32"}}
|> ProductFilter.params_to_scopes()
|> ProductQuery.scope()
|> Repo.all()



  

MIT License
Copyright (c) 2021 David Librera
Permission is hereby granted, free of charge, to any person obtaining
a copy of this software and associated documentation files (the
"Software"), to deal in the Software without restriction, including
without limitation the rights to use, copy, modify, merge, publish,
distribute, sublicense, and/or sell copies of the Software, and to
permit persons to whom the Software is furnished to do so, subject to
the following conditions:
The above copyright notice and this permission notice shall be
included in all copies or substantial portions of the Software.
THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF
MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE
LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION
OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION
WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.

An opinable set of proposed patters that helps to write reusable and no
repetitive code for Contexts.
Motivation
After several years in Ruby on Rails developing I've
got used to structure code folllowing the Single Responsibility Principle.
Phoenix comes with the
Context concept, a module that cares
about expose an API of an application section to other sections.
In a Context we usually have at least 6 actions for each defined schema
(list_records/0, get_record!/1, create_record/1, update_record/2,
delete_record/1, change_record/2).
If we consider that all Business Logic could go inside the Context, it's possibile
to have a module with hundreds of lines of code, making code mainteinance very
hard to be guardanteed.
The idea is to highlight common portion of code that can be extacted and moved
into a separated module with only one responsibility and that could be reused
in different contexts.
List records
The method list_* has a default implementation that returns  the list of
associated record:
def list_records do
  MyApp.Context.RecordSchema
  |> MyApp.Repo.all()
end
Let's add now to the context the capability of filtering the collection according
to an arbitrary set of scopes, calling the function in this way:
iex> Context.list_records(title_matches: "value")
A possible solution could be to delegate the query building into a separated
RecordQuery module
defmodule RecordQuery do
  def scope(list_of_filters) do
    RecordSchema
    |> improve_query_with_filters(list_of_filters)
  end

  defp improve_query_with_filters(start_query, list_of_filters) do
    list_of_filters
    |> Enum.reduce(start_query, fn scope, query -> apply_scope(query, scope) end)
  end

  defp apply_scope(query, {:title_matches, "value"}) do
    query
    |> where([q], ...)
  end

  defp apply_scope(query, {:price_lte, value}) do
    query
    |> where([q], ...)
  end
end
and use it into the Context
def list_records(scopes \\ []) do
  RecordQuery.scope(scopes)
  |> Repo.all()
end

iex> Context.list_records(title_matches: "value", price_lte: 42)
PhoenixBricks.Scopes
Using PhoenixBricks.Scopes it's possible to extend a module with all scope
behaviours:
defmodule RecordQuery do
  use PhoenixBricks.Scopes, schema: RecordSchema

  defp apply_scope(query, {:title_matches, "value"}) do
    query
    |> where([q], ...)
  end
end
Filter
Another common feature is to filter records according to params provided through
url params (for example after a submit in a search form).
def index(conn, params)
  filters = Map.get(params, "filters", %{})

  colletion = Context.list_records_based_on_filters(filters)

  conn
  |> assign(:collection, collection)
  ...
end
ensuring to allow only specified filters
A possible implementation could be:
defmodule RecordFilter do
  @search_filters ["title_matches", "price_lte"]

  def convert_filters_to_scopes(filters) do
    filters
    |> Enum.map(fn {name, value} ->
      convert_filter_to_scope(name, value)
    end)
  end

  def convert_filter_to_scope(name, value) when name in @search_fields do
    {String.to_atom(name), value}
  end
end
This way parameters are filtered and converted to a Keyword that is the common
format for the RecordQuery described above.
iex> RecordFilter.convert_filters_to_scopes(%{"title_matches" => "value", "invalid_scope" => "value"})
iex> [title_matches: "value"]
and we can rewrite the previous action emphasizing the params convertion and
the collection filter
def index(conn, params) do
  filters = Map.get(params, "filters", %{})

  collection =
    filters
    |> RecordFilter.convert_filters_to_scopes()
    |> Context.list_records()

  conn
  |> assign(:collection, collection)
  ....
end
The last part is to build a search form. In order to achieve this, we can add
schema functionality to RecordFilter module:
defmodule RecordFilter do
  use Ecto.Schema

  embedded_schema do
    field :title_matches, :string
  end

  def changeset(filter, params) do
    filter
    |> cast(params, [:title_matches])
  end
end

def index(conn, params) do
  filters = Map.get(params, "filters", %{})
  filter_changeset = RecordFilter.changeset(%RecordFilter{}, filters)

  collection =
    filters
    |> RecordFilter.convert_filters_to_scopes()
    |> Context.list_records()

  conn
  |> assign(:collection, collection)
  |> assign(:filter_changeset, filter_changeset)
end
  <%= f = form_for @filter_changeset, .... %>
    <%= label f, :title_matches %>
    <%= text_input f, :title_matches %>

    <%= submit "Filter results" %>
  <% end %>
PhoenixBricks.Filter
Using PhoenixBricks.Filter module it's possible to extend a module with all
filtering behaviours (define a changeset and the filter convertion)
defmodule RecordFilter do
  use PhoenixBricks.Filter,
      filters: [
        title_matches: :string
      ]
end
making available changeset/2 defined above and convert_filters_to_scopes/1

Defines a Filter Schema
Examples
defmodule RecordFilter do
  use PhoenixBricks.Filter,
      filters: [
        field_matcher: :string
      ]
end
It defines a schema which fields could be used in a search form.
Search changeset
def index(conn, params) do
  filters = Map.get(params, "filters", %{})

  conn
  |> assign(:changeset, RecordFilter.changeset(%RecordFilter{}, filters))
  |> render("index.html")
end
<%= form_for @conn, Routes.session_path(@conn, :create), [method: :post, as: :user], fn f -> %>
  <div class="form-group">
    <%= label f, :field_matcher %>
    <%= text_input f, :field_matcher %>
  </div>

  <div class="form-group">kab
    <%= submit "Search" %>
  </div>
<% end %>
Convertion from map of params to a list of filtered scopes
iex> params = %{"filters" => %{"field_matcher" => "value"}}
iex> RecordFilter.convert_params_to_scopes(params)
iex> [field_matcher: "value"]

Defines a common interface for adding scopes to a Schema
Examples
defmodule RecordQuery do
  use ExBricks.Scopes, schema: Record
end
It provides you a method scope that improve the empty scope of provided schema
with additional scopes.
scope/0 returns the empty scope
iex> RecordQuery.scope()
iex> Record
With scope/1 you can provide a list of scopes that improve the starting scope
iex> scopes = [field1: {:eq, "value"}, field2: {:gte, 42}]
iex> RecordQuery.scope(scopes)
iex> #Ecto.Query<from r0 in Record, where: r0.field1 == ^"value" and r0.field2 >= 42>
If you need to improve an existing Ecto.Query you can use scope/2
iex> starting_scope = from(r in Record, where: r.field1 == "value")
iex> scopes = [field2: {:gte, 42}]
iex> RecordQuery.scope(starting_scope, scopes)
iex> #Ecto.Query<from r0 in Record, where: r0.field1 == ^"value" and r0.field2 >= 42>
Built-in scopes
:eq
iex> RecordQuery.scope(field: {:eq, "value"})
iex> #Ecto.Query<from r0 in Record, where: r0.fields == ^"value">
:gt
iex> RecordQuery.scope(field: {:gt, "value"})
iex> #Ecto.Query<from r0 in Record, where: r0.fields > ^"value">
:lt
iex> RecordQuery.scope(field: {:lt, "value"})
iex> #Ecto.Query<from r0 in Record, where: r0.fields < ^"value">
:gte
iex> RecordQuery.scope(field: {:gte, "value"})
iex> #Ecto.Query<from r0 in Record, where: r0.fields >= ^"value">
:lte
iex> RecordQuery.scope(field: {:lte, "value"})
iex> #Ecto.Query<from r0 in Record, where: r0.fields <= ^"value">
:neq
iex> RecordQuery.scope(field: {:neq, "value"})
iex> #Ecto.Query<from r0 in Record, where: r0.fields != ^"value">
:matches
iex> RecordQuery.scope(field: {:matches, "value"})
iex> #Ecto.Query<from r0 in Record, where: ilike(r0.field, ^"%value%")>
Customize scopes
If you need to define more comprehensive scopes you can improve the query adding
new apply_scope/2 methods
defmodule RecordQuery, do
  use ExBricks.Scopes, schema: Record

  defp apply_scope(query, {:name_matches, "value"}) do
    query
    |> apply_scope(:name, {:matches, "value"})
  end

  defp apply_scope(query, {:published}) do
    query
    |> where([p], p.status == "published")
  end
end

iex> RecordQuery.scope(:published, name_matches: "value")
iex> #Ecto.Query<from r0 in Record, where: r0.status == "published" and ilike(r0.name, ^"%value%")>

Generates a Filter schema around an Ecto schema

      





  

    
mix phx.bricks.gen.query
    



      
Generates a Query module that improves an Ecto.Query of provided schema

